



Legac-E Education

Print members of a PDS or PDSE - COBOL

In 2018 I produced a COBOL program to list the contents of a PDS or PDSE directory. The code here, produced over two days in October 2021 on a z/OS V2.R4 system, expands the concept to enable printing of selected members from a PDS or PDSE. As coded the program is designed to work with text libraries, i.e. program source libraries, JCL procedure libraries and parameter libraries.

The program uses BPXWDYN to perform the dynamic allocation of the PDS or PDSE avoiding the need for DD statements, which is relevant as it does two allocations for the same data set, one for the directory and the other to access the members.

It also uses the Language Environment (LE) module CEE3ABD to issue a User Abend if the data set name provided is deemed to be invalid.

The program also calls IDCAMS to ensure that the data set selected exists in the catalogue.

Further documentation is provided as comments within the source code.



Legac-E Education

Print members of a PDS or PDSE - COBOL

```
CBL MAP,DYN
IDENTIFICATION DIVISION.
PROGRAM-ID.                                PRINTPDS.
*-----*
*   This program is designed to print members of a PDS/PDSE   *
*   provided the members have a record size of 80 bytes.     *
*-----*
*   Primary Input - DD Statement MEMBERS                      *
*   The first record will be the PDS/PDSE data set name (DSN) *
*   Subsequent records will contains member names in        *
*   columns 1 to 8.                                         *
*-----*
*   Primary Output - DD Statement PDSPRINT                   *
*   This will be a series of 121 byte records displaying the *
*   contents of a member, preceded by a title line showing  *
*   the member name.                                       *
*-----*
*   Secondary Output - DD Statement SYSOUT                   *
*   This will be a series of 121 byte records containing     *
*   both the SORT output and messages indicating any members *
*   not found in the PDS/PDSE.                             *
*-----*
*   Error Conditions                                         *
*   U0240 - Indicates that the first record of the MEMBERS  *
*   data set does not conform to data set naming conventions.*
*   Any U4nnn ABENDS will be produced by LE.               *
*   U0250 - Indicates that the specified data set name is   *
*   not catalogued.                                         *
*-----*
*   PDS/PDSE Handling.                                       *
*   This is done via two dynamically allocated DD statements *
*   LIBRARY - Provides access to the file directory         *
*   PDSLIB  - Is used to gain access to individual members *
*-----*
ENVIRONMENT DIVISION.
INPUT-OUTPUT SECTION.
FILE-CONTROL.
    SELECT DIRECTORY-IN          ASSIGN TO UT-S-LIBRARY.
    SELECT IDCAMS-IN            ASSIGN TO UT-S-SYSIN.
    SELECT LIBRARY-IN           ASSIGN TO UT-S-PDSLIB.
    SELECT MEMBERS-IN           ASSIGN TO UT-S-MEMBERS.
    SELECT PRINT-FILE           ASSIGN TO UT-S-PDSPRINT.
    SELECT SORT-FILE            ASSIGN TO UT-S-SORTWK01.
DATA DIVISION.
FILE SECTION.
```



Legac-E Education

Print members of a PDS or PDSE - COBOL

```
FD DIRECTORY-IN                RECORDING MODE IS F
                                BLOCK CONTAINS 0 RECORDS
                                RECORD CONTAINS 256 CHARACTERS
                                DATA RECORD IS FD-DIR-BLOCK.
01 FD-DIR-BLOCK                PIC X(256).
FD IDCAMS-IN                   RECORDING MODE IS F
                                BLOCK CONTAINS 1 RECORDS
                                RECORD CONTAINS 80 CHARACTERS
                                DATA RECORD IS FD-AMS-COMMAND.
01 FD-AMS-COMMAND              PIC X(80).
FD LIBRARY-IN                  RECORDING MODE IS F
                                BLOCK CONTAINS 0 RECORDS
                                RECORD CONTAINS 80 CHARACTERS
                                DATA RECORD IS FD-PDSREC.
01 FD-PDSREC                   PIC X(80).
FD MEMBERS-IN                  RECORDING MODE IS F
                                BLOCK CONTAINS 0 RECORDS
                                RECORD CONTAINS 80 CHARACTERS
                                DATA RECORD IS FD-INMEM.
01 FD-INMEM                     PIC X(80).
FD PRINT-FILE                   RECORDING MODE IS F
                                BLOCK CONTAINS 0 RECORDS
                                RECORD CONTAINS 121 CHARACTERS
                                DATA RECORD IS FD-PLINE.
01 FD-PLINE                     PIC X(121).
SD SORT-FILE                   DATA RECORD IS SD-INPUT.
01 SD-INPUT.
    03 SD-MEMBER.
        05 SD-MCHAR              PIC X OCCURS 8 TIMES.
    03                          PIC X(72).
WORKING-STORAGE SECTION.
01 ADHOC-BINARY-FIELDS.
    03 ABEND-CODE                PIC S9(8) COMP VALUE +240.
    03 AMS-IOLIST                PIC S9(8) COMP VALUE 0.
    03 REASON-CODE               PIC S9(8) COMP VALUE +0.
    03 AMS-AUXLIST               PIC S9(4) COMP VALUE +0.
    03 AMS-OPTIONS               PIC S9(4) COMP VALUE +0.
    03 AMS-PAGENO                PIC S9(4) COMP VALUE +0.
    03 DBLK-SIZE                 PIC S9(4) COMP VALUE +0.
    03 DIR-COUNT1                PIC S9(4) COMP VALUE +0.
*   DIR-COUNT1 Tally of member processing passes.
    03 DIR-COUNT2                PIC S9(4) COMP VALUE +0.
*   DIR-COUNT2 number of entries in a directory block.
    03 DSN-COUNT                 PIC S9(4) COMP VALUE 0
                                OCCURS 5 TIMES.
    03 DSN-PTR                   PIC S9(4) COMP VALUE +1.
    03 DSN-LENGTH                PIC S9(4) COMP VALUE +1.
```



Legac-E Education

Print members of a PDS or PDSE - COBOL

```
03 MEM-LENGTH          PIC S9(4) COMP VALUE +1.
03 WS-CALC              PIC S9(4) COMP VALUE +0.
03                      REDEFINES WS-CALC.
    05                  PIC X.
    05 D-REST           PIC X.
03 WS-CNT              PIC S9(4) COMP VALUE +0.
03 WS-DPOS             PIC S9(4) COMP VALUE +1.
03 WS-PERIODS         PIC S9(4) COMP VALUE +0.
03 WS-SUB             PIC S9(4) COMP VALUE +1.
01 ADHOC-NON-BINARY-FIELDS.
03 DIR-FLAG           PIC X VALUE SPACE.
    88 DIR-EXHAUSTED  VALUE 'D'.
03 IDC-AMS            PIC X(8) VALUE 'IDCAMS'.
03 LINE-COUNT        PIC S999 COMP-3 VALUE +58.
03 MEM-IN-FLAG       PIC X VALUE SPACE.
    88 END-OF-FILE    VALUE 'D'.
03 MEM-IO-FLAG       PIC X VALUE SPACE.
    88 END-OF-MEMBER  VALUE 'D'.
03 NOT-FOUND-MSG.
    05                PIC X VALUE SPACE.
    05 NFM-MEMBER    PIC X(8).
    05                PIC X(112) VALUE
' Not found in library.'.
03 PAGE-COUNT        PIC S999 COMP-3 VALUE 0.
03 SORT-FLAG         PIC X VALUE SPACE.
    88 SORT-END      VALUE 'D'.
03 DSN-REC.
    05 DSN-CHARS     OCCURS 44 TIMES.
        07 DSN-RCHAR PIC X.
    05                PIC X(36).
03 DATA-SET-NAME    REDEFINES DSN-REC.
    05 DS-NAME       PIC X(44).
    05                PIC X(36).
01 WS-AMS-COMMAND.
03                      PIC X(11) VALUE ' LISTC ENT('.
03 AMS-DSN           PIC X(50) VALUE SPACES.
03                      PIC X(19) VALUE SPACES.
01 NAME-TABLE.
03 DIR-NAMES         OCCURS 21 TIMES
                      INDEXED BY WS-INDX.
    05 MEMBER-NAME    PIC X(8).
01 NAME-LIST         REDEFINES NAME-TABLE.
03                      PIC X(168).
```



Legac-E Education

Print members of a PDS or PDSE - COBOL

```
01  INVALID-DSN.
    03          PIC X VALUE SPACE.
    03          PIC X(33) VALUE
        'PRINTPDS - Invalid data set name '.
    03  BAD-DSN  PIC X(44).
    03          PIC X(47) VALUE SPACES.
01  DSN-NOT-FOUND.
    03          PIC X VALUE SPACE.
    03          PIC X(42) VALUE
        'PRINTPDS - Data set name not catalogued : '.
    03  NOT-CAT  PIC X(44).
    03          PIC X(35) VALUE SPACES.
01  DSN-NODES    PIC X(45) VALUE SPACES.
01  DSN-NODE-TAB REDEFINES DSN-NODES.
    03  DSN-ELEMENT PIC X(9) OCCURS 5 TIMES.
01  PRINT-LINE.
    03          PIC X VALUE SPACE.
    03  PRT-PDSREC  PIC X(80).
    03          PIC X(40) VALUE SPACES.
01  TITLE-LINE.
    03          PIC X VALUE SPACE.
    03          PIC X(10) VALUE SPACES.
    03  TITLE-TEXT  PIC X(80) VALUE SPACES.
    03  PAGE-NO     PIC ZZ9.
    03          PIC X(30) VALUE SPACES.
01  ALLOC-DIRECTORY.
    03          PIC S9(4) COMP VALUE +118.
    03          PIC X(47) VALUE
        'ALLOC FI(LIBRARY SHR MSG(2) LRECL(256) RECFM(F)'.
    03          PIC X(16) VALUE
        'BLKSIZE(256) DA('.
    03  DIR-DSN    PIC X(55) VALUE SPACES.
01  ALLOC-PDSLIB.
    03          PIC S9(4) COMP VALUE +86.
    03          PIC X(31) VALUE
        'ALLOC FI(PDSLIB) SHR MSG(2) DA('.
    03  PDS-DSN   PIC X(55) VALUE SPACES.
01  DEALLOC-PDSLIB.
    03          PIC S9(4) COMP VALUE +16.
    03          PIC X(16) VALUE
        'FREE FI(PDSLIB) '.
01  AMS-DDNAMES.
    03          PIC S9(4) COMP VALUE 0.
    03          PIC S9(8) COMP VALUE 0.
    03          PIC S9(8) COMP VALUE 0.
    03          PIC S9(8) COMP VALUE 0.
    03          PIC S9(8) COMP VALUE 0.
```



Legac-E Education

Print members of a PDS or PDSE - COBOL

```

      03                                PIC X(8) VALUE 'AMSIN'.
      03                                PIC X(8) VALUE 'AMSPRINT'.
01  WS-DIRBLOCK.
      03  ACTBLK                        PIC S9(4) COMP.
      03  MEMLIST                       PIC X(254).
-----*
*   Individual basic directory entry is 12 bytes as per:      *
*       05  MEMNAME                    PIC X(8).              *
*       05  TTR                        PIC X(3).              *
*       05  STAT-CNT                   PIC X.                  *
*   This last field contains three pieces of information, but *
*   the last five bits are of interest as they hold the number *
*   of additional half-words in the directory entry.          *
*-----*
*   If the member name is HIGH-VALUES X'FF' then the end of   *
*   the member list has been reached.                          *
*   Because the individual directory entries are unlikely to   *
*   fill the entire directory block, be aware that unused     *
*   bytes will contain LOW-VALUES X'00'.                       *
*-----*
LINKAGE SECTION.
PROCEDURE DIVISION.
A010-MAINLINE SECTION.
-----*
*   This is the logical start of program PRINTPDS.             *
*-----*
      OPEN                                INPUT MEMBERS-IN
                                           OUTPUT PRINT-FILE.

      PERFORM B030-VALIDATE-DSN.
      CALL 'BPXWDYN'                      USING ALLOC-DIRECTORY.
      OPEN                                INPUT DIRECTORY-IN.
-----*
*   Remember PDS directories have the member names stored in   *
*   collating sequence.                                        *
*-----*
      SORT SORT-FILE                      ON ASCENDING KEY
                                           SD-MEMBER
                                           INPUT PROCEDURE B010-GET-NAME
                                           OUTPUT PROCEDURE B020-GET-MEMBER.

      CLOSE                                MEMBERS-IN
                                           DIRECTORY-IN
                                           PRINT-FILE.
-----*
*   This is the logical end of program PRINTPDS.              *
*-----*
A010-MAINLINE-END.
      GOBACK.
```



Legac-E Education

Print members of a PDS or PDSE - COBOL

```
B010-GET-NAME SECTION.
*-----*
*   This section of the program reads the list of members to   *
*   be printed and passes them to the SORT.                   *
*-----*
      PERFORM                               UNTIL END-OF-FILE
        READ MEMBERS-IN                     AT END MOVE 'D' TO MEM-IN-FLAG
        END-READ
        IF NOT END-OF-FILE                   RELEASE SD-INPUT
                                           FROM FD-INMEM

      END-IF
      END-PERFORM.
B010-GET-NAME-EXIT.
      EXIT.
B020-GET-MEMBER SECTION.
*-----*
*   This section of the program retrieves the sorted records,  *
*   checks if the member exists, and prints it if it does.    *
*-----*
      PERFORM C010-GET-DIRECTORY-BLOCK.
      PERFORM                               UNTIL SORT-END
                                           OR DIR-EXHAUSTED
      RETURN SORT-FILE RECORD               AT END MOVE 'D' TO SORT-FLAG
      END-RETURN
      IF SD-MEMBER > MEMBER-NAME (DIR-COUNT2)
        PERFORM UNTIL SD-MEMBER <=      MEMBER-NAME (dir-count2)
                                           OR DIR-EXHAUSTED
          PERFORM C010-GET-DIRECTORY-BLOCK
          END-PERFORM
        END-IF
        SET WS-INDX                       TO 1
        SEARCH DIR-NAMES                   VARYING WS-INDX
          AT END
                                           MOVE SD-MEMBER TO NFM-MEMBER
                                           DISPLAY NOT-FOUND-MSG
                                           UPON SYSOUT
          WHEN MEMBER-NAME (WS-INDX) = SD-MEMBER
            PERFORM C020-PRINT-MEMBER

      END-SEARCH
      END-PERFORM.
B020-GET-MEMBER-EXIT.
      EXIT.
```



Legac-E Education

Print members of a PDS or PDSE - COBOL

```
B030-VALIDATE-DSN SECTION.
*-----*
* Retrieve and process the data set (library) name. *
*-----*

READ MEMBERS-IN          INTO DATA-SET-NAME
                          AT END MOVE 'D' TO MEM-IN-FLAG

END-READ
INSPECT DS-NAME          TALLYING WS-PERIODS
                          FOR ALL "." BEFORE SPACE.

IF WS-PERIODS NOT > 0
  MOVE DS-NAME           TO BAD-DSN
  DISPLAY INVALID-DSN   UPON SYSOUT
  CALL 'CEE3ABD'        USING ABEND-CODE, REASON-CODE
END-IF.
ADD +1                  TO WS-PERIODS.
PERFORM                 WS-PERIODS TIMES
  UNSTRING DS-NAME      DELIMITED BY '.'
                        OR ALL ' ' INTO
                        DSN-ELEMENT (WS-SUB)
                        COUNT IN DSN-COUNT (WS-SUB)
                        WITH POINTER DSN-PTR

  END-UNSTRING
  ADD +1                 TO WS-SUB
END-PERFORM.
MOVE +1                 TO WS-SUB.
PERFORM                 WS-PERIODS TIMES
  IF DSN-COUNT (WS-SUB) > 8
    MOVE DS-NAME        TO BAD-DSN
    DISPLAY INVALID-DSN UPON SYSOUT
    CALL 'CEE3ABD'     USING ABEND-CODE, REASON-CODE
  END-IF
  ADD +1                 TO WS-SUB
END-PERFORM.
PERFORM UNTIL           DSN-RCHAR (DSN-LENGTH) = ' '
  ADD +1                 TO DSN-LENGTH
END-PERFORM.
SUBTRACT +1             FROM DSN-LENGTH.
MOVE DS-NAME(1:DSN-LENGTH) TO DIR-DSN.
STRING DS-NAME(1:DSN-LENGTH) ' '
                        DELIMITED BY SIZE
                        INTO AMS-DSN.
OPEN                    OUTPUT IDCAMS-IN.
WRITE FD-AMS-COMMAND   FROM WS-AMS-COMMAND.
CLOSE                   IDCAMS-IN.
CALL IDC-AMS           USING AMS-OPTIONS
                        AMS-DDNAMES
                        AMS-PAGENO
                        AMS-IOLIST
```




Legac-E Education

Print members of a PDS or PDSE - COBOL

```

                                AMS-AUXLIST.
IF RETURN-CODE > 0
    MOVE DS-NAME(1:DSN-LENGTH) TO NOT-CAT
    DISPLAY DSN-NOT-FOUND UPON SYSOUT
    MOVE +250                    TO ABEND-CODE
    CALL 'CEE3ABD'                USING ABEND-CODE, REASON-CODE
END-IF.
B030-VALIDATE-DSN-EXIT.
EXIT.
C010-GET-DIRECTORY-BLOCK SECTION.
*-----*
*   Build the NAME-LIST from the current Directory Block.   *
*-----*
    MOVE 0                        TO DIR-COUNT2.
    MOVE SPACES                   TO NAME-LIST.
    READ DIRECTORY-IN             INTO WS-DIRBLOCK
                                AT END
                                MOVE 'D' TO DIR-FLAG

    END-READ
    MOVE +1                       TO WS-DPOS
    COMPUTE DBLK-SIZE =           (ACTBLK - 2)
    IF ACTBLK NOT = +14
        PERFORM                   UNTIL WS-DPOS > DBLK-SIZE
*-----*
*   Extract each directory entry from the current block.   *
*-----*
        IF MEMLIST(WS-DPOS:8) NOT = HIGH-VALUES
            ADD +1                 TO DIR-COUNT2
            MOVE MEMLIST(WS-DPOS:8) TO MEMBER-NAME (DIR-COUNT2)
            COMPUTE WS-CNT          = (WS-DPOS + 11)
            MOVE MEMLIST(WS-CNT:1)  TO D-REST
*-----*
*   The following three IF instructions purify the user count *
*   field by removing the HIGH-ORDER three bits if set.   *
*-----*
            IF WS-CALC > 127        SUBTRACT 128 FROM WS-CALC
            END-IF
            IF WS-CALC > 63         SUBTRACT 64 FROM WS-CALC
            END-IF
            IF WS-CALC > 31         SUBTRACT 32 FROM WS-CALC
            END-IF
            COMPUTE WS-DPOS =       (WS-CALC * 2) + WS-CNT + 1
        ELSE
            MOVE +257 TO WS-DPOS
        END-IF
    END-PERFORM.
    MOVE +1                        TO DIR-COUNT1.
```



Legac-E Education

Print members of a PDS or PDSE - COBOL

```
C010-GET-DIRECTORY-BLOCK-EXIT.
  EXIT.
C020-PRINT-MEMBER SECTION.
*-----*
*   This section of the program produces the actual print of   *
*   the selected member.                                       *
*-----*

  MOVE +1                                TO MEM-LENGTH.
  PERFORM UNTIL                          SD-MCHAR (MEM-LENGTH) = ' '
    ADD +1                                TO MEM-LENGTH
  END-PERFORM.
  SUBTRACT +1                            FROM MEM-LENGTH.
  MOVE SPACES                             TO PDS-DSN.
  STRING DS-NAME(1:DSN-LENGTH) ' (' SD-MEMBER(1:MEM-LENGTH)
                                ') ' DELIMITED BY SIZE
                                INTO PDS-DSN.
  CALL 'BPXWDYN'                          USING ALLOC-PDSLIB.
  OPEN                                    INPUT LIBRARY-IN.
  MOVE ' '                                TO MEM-IO-FLAG.
  MOVE +58                                TO LINE-COUNT.
  PERFORM UNTIL END-OF-MEMBER
    READ LIBRARY-IN                       INTO PRT-PDSREC
    AT END MOVE 'D' TO MEM-IO-FLAG

  END-READ
  IF NOT END-OF-MEMBER
    IF LINE-COUNT > +57
      MOVE +0 TO LINE-COUNT
      ADD +1 TO PAGE-COUNT
      MOVE PAGE-COUNT TO PAGE-NO
      STRING 'Print of member : '
            SD-MEMBER
            DELIMITED BY SIZE
            INTO TITLE-TEXT
      END-STRING
      WRITE FD-PLINE
            FROM TITLE-LINE

    END-IF
    WRITE FD-PLINE                        FROM PRINT-LINE
    ADD +1                                TO LINE-COUNT
  END-IF
  END-PERFORM.
  CLOSE                                    LIBRARY-IN.
  CALL 'BPXWDYN'                          USING DEALLOC-PDSLIB.
C010-PRINT-MEMBER-EXIT.
  EXIT.
*-----*
*   This is the physical end of program PRINTPDS.             *
*-----*
```



Legac-E Education

Print members of a PDS or PDSE - COBOL

Invoking JCL

```
//S0010      EXEC  PGM=PRINTPDS
//STEPLIB   DD    DISP=SHR,DSN=&SYSUID..LINKLIB
//SYSOUT    DD    SYSOUT=*
//SYSPRINT  DD    DUMMY
//PDSPRINT  DD    SYSOUT=*
//SYSIN     DD    UNIT=SYSDA,SPACE=(TRK,1)
//MEMBERS   DD    *,SYMBOLS=EXECSYS
&SYSUID..SOURCE.COB2
PRINTPDS
GETDSN
```